GRAPHES 2D : SOLUTIONS DES EXERCICES

Bernard Dupont

Bernard.Dupont@univ-lille1.fr

Exercice M.1

Enoncé

Joindre par des segments de doite les points du plan de coordonnées $(1;4)$, $(-2;-3)$, $(4;-5)$ et $(-6;5)$. Les points sont des petits cercles de couleur noire et les segments de couleur bleue.

Solution

La commande **plot** permet de tracer des points dans le plan grâce à l'option **style=point** et des droites joignant des points successifs avec l'option **style=line**.On commence par construire un graphique labellisé pour les points et un graphique labellisé pour les segments. Chacun contient en options les directives graphiques.

```
> restart;
  G1:=plot([[1,4],[-2,-3],[4,-5],[-6,5]],x=-7..7,y=-7..7,
  style=point,color=black,symbol=circle):
  G2:=plot([[1,4],[-2,-3],[4,-5],[-6,5]],x=-7..7,y=-7..7,
  style=line,color=blue):
```

Il ne reste plus qu'à charger le paquetage **plots** et appeler la commande **display**.

```
> with(plots):
  display(G1,G2);
```



Exercice M.2

Enoncé

L'option **`filled=true`** de la commande plot permet de colorer la surface comprise entre la courbe représentative d'une fonction et l'axe des abscisses. A l'aide de la commande **`display`** du paquetage **`plots`**, construire un graphique colorant en rouge la surface comprise entre la courbe représentative de $y=\cos(x)$ et celle de $y=\sin(x)$ sur l'intervalle $[0;4\pi]$.

Solution

Cet exercice demande réflexion. Une chose est sûre : pour tout x de l'intervalle $[0;4\pi]$, seuls les points y vérifiant $\min(\cos(x),\sin(x)) \le y \le \max(\cos(x),\sin(x))$ seront de couleur rouge alors que ceux vérifiant $y \le \min(\cos(x),\sin(x))$ et $y \ge \max(\cos(x),\sin(x))$ seront colorés en blanc.

On commence par tracer les courbes représentatives des deux fonctions ainsi que les surfaces comprises entre les courbes et l'axe des abscisses.

```
> restart;
  gsin:=plot(sin,0..4*Pi,color=black):
  sinus:=plot(sin,0..4*Pi,filled=true):
  gcos:=plot(cos,0..4*Pi,color=black):
  cosinus:=plot(cos,0..4*Pi,filled=true):
```

Ensuite, on construit une fonction-procédure chargée de colorier en blanc les zones "extérieures" :

```
> f:=x->if cos(x)>0 and sin(x)>0 then min(cos(x),sin(x))
  elif cos(x)<0 and sin(x)<0 then max(cos(x),sin(x))
  else 0
  end if:
  graf:=plot(f,0..4*Pi,filled=true,color=white):
```

Enfin, on réunit les graphiques avec **`display`** en respectant l'ordre impératif suivant : colorisation en blanc des surfaces "extérieures", tracé des courbes puis coloration des surfaces.

```
> with(plots):
  display([graf,gsin,gcos,sinus,cosinus]);
```



Exercice M.3

Enoncé

Soit la fonction f définie par $f(x) = \sqrt[3]{x^3 - 3x + 2}$.

1. Donner une représentation graphique pertinente.

2. Déterminer ses limites en $+\infty$ et $-\infty$.

3. Etudier la dérivabilité de f en $x = -2$ et $x = 1$.

4. Montrez que f admet un maximum local en $x = -1$.

Solution

1. A première vue, il suffit de définir la fonction puis de la représenter en utilisant directement **plot** puisque son domaine de définition est $\mathbb{R}$. Malheureusement, ça ne marche pas :

```
> restart:
  f:=x->(x^3-3*x+2)^(1/3);
  plot(f(x),x=-5..5);
```

$$f := x \rightarrow \left(x^3 - 3x + 2\right)^{1/3}$$



Maple ne donne aucune image aux abscisses inférieures à -2. D'ailleurs, un message d'avertissement est déclenché si on s'obstine à vouloir une représentation graphique sur tout intervalle situé à gauche de -2.

```
> plot(f(x),x=-10..-4);
Warning, unable to evaluate the function to numeric values
in the region; see the plotting command's help page to
ensure the calling sequence is correct
```



Qu'est ce à dire? En fait, le logiciel bascule dans le corps des complexes à gauche de -2 et ne peut donc pas représenter les images dans le plan réel *xOy*.

```
> evalf(f(-3));
```

$$1.259921050 + 2.182247272\,I \qquad \textbf{(3.1)}$$

Il faut ruser en créant une fonction définie par morceaux. A droite de -2, on garde la fonction donnée dans l'énoncé. A gauche, on "sort" le signe "moins" de l'évaluation de la racine troisième.

```
> restart:
  f:=x->piecewise(x>=-2,(x^3-3*x+2)^(1/3),-(abs(x)^3-3*abs(x)
  -2)^(1/3));
  evalf(f(5));evalf(f(-5));#vérification
```

$$f := x \rightarrow piecewise\left(-2 \leq x,\ \left(x^3 - 3\,x + 2\right)^{1/3},\ -\left(|x|^3 - 3\,|x| - 2\right)^{1/3}\right)$$
$$4.820284528$$
$$-4.762203156 \qquad \textbf{(3.2)}$$

Cette fois, on "tient" notre graphe.

```
> plot(f(x),x=-5..5,thickness=2,caption=typeset
  ("Représentation graphique de ",'f(x)'=(x^3-3*x+2)^(1/3)));
```



Représentation graphique de
$f(x) = (x^3 - 3\,x + 2)^{1/3}$

2.Le calcul de ces deux limites ne pose aucun problème à la commande **limit**.

```
> limit(f(x),x=infinity);#limite en +∞
```

$$\infty \tag{3.3}$$

```
> limit(f(x),x=-infinity);#limite en -∞
```

$$-\infty \tag{3.4}$$

3.Pour le plaisir, demandons la fonction dérivée première de la fonction en appliquant la commande **D** à **f**.

```
> fp:=D(f);
```

$$fp := x \rightarrow \mathit{piecewise}\left(x < -2,\ -\frac{1}{3}\,\frac{-3\,x^2+3}{(-2-x^3+3\,x)^{2/3}},\ x=-2,\ \infty,\ -2 < x,\ \frac{1}{3}\,\frac{3\,x^2-3}{(x^3-3\,x+2)^{2/3}}\right) \tag{3.5}$$

En fait, l'output indique deux choses :

• Pour tout x différent de -2, la dérivée est $f'(x) = \frac{1}{3}\,\frac{3\,x^2-3}{(x^3-3\,x+2)^{2/3}}$ (quand x est inférieur à -2, l'expression renvoyée par Maple est équivalente).
• Quand $x = -2$, la dérivée vaut l'infini (ce que confirme le graphique de départ). Rigoureusement parlant, f n'est pas dérivable en ce point.
En revanche, Maple ne voit pas de problème en $x = 1$ alors que le dénominateur de la fonction rationnelle f' s'annule pour cette valeur.

```
> fp(1);
Error, (in fp) numeric exception: division by zero
```

Pour connaître ce qui se passe en ce point, il faut repasser par la définition d'une dérivée.

```
> limit((f(1+h)-f(1))/h,h=0,right);#dérivée à droite
  limit((f(1+h)-f(1))/h,h=0,left);#dérivée à gauche
```

$$\infty$$
$$-\infty \tag{3.6}$$

On en déduit que f n'est pas dérivable en $x = 1$ et présente un point de rebroussement en ce point (ce que confirme le graphique de départ).

4. Cherchons les points candidats éventuels par annulation de la dérivée première.

```
> solve(fp(x)=0,x);
```

$$-1 \tag{3.7}$$

On a un point candidat pour lequel on calcule la valeur prise par la dérivée seconde de f.

```
> D(D(f))(-1);is(%<0);
```

$$-\frac{1}{2}\,4^{1/3}$$

$$\mathit{true} \tag{3.8}$$

Comme $f''(-1) < 0$, f présente un maximum local en ce point (encore une fois : ce que confirme le graphe).

Exercice E.1

Enoncé

Soit la fonction de production Cobb-Douglas $Q = L^{\frac{2}{3}} K^{\frac{1}{3}}$. Tracer dans le même plan (quantité de travail, quantité de capital) = (L, K) les isoquants de production $Q = 10$, $Q = 20$, $Q = 30$.

Solution

Il s'agit de tracer trois courbes implicites dans le même graphique. Une utilisation naïve de Maple donne :

```
> restart;
  with(plots):
  cobb_douglas:=l^(2/3)*k^(1/3)-q=0:
  co10:=subs(q=10,cobb_douglas):
  g1:=implicitplot(co10,l=0.01..10,k=1..500,grid=[100,100]):
  co20:=subs(q=20,cobb_douglas):
  g2:=implicitplot(co20,l=0.01..10,k=1..500,grid=[100,100]):
  co30:=subs(q=30,cobb_douglas):
  g3:=implicitplot(co30,l=0.01..10,k=1..500,grid=[100,100]):
  display([g1,g2,g3]);
```



Mais on peut notablement accélérer le travail en définissant tout d'abord une fonction dépendant d'un paramètre :

```
> F:=q->l^(2/3)*k^(1/3)-q:
  implicitplot({F(10),F(20),F(30)},l=0.01..10,k=1..500,grid=
  [100,100]);
```



Exercice E.2

Enoncé

Cet exercice porte sur le modèle IS-LM simple. L'équation d'équilibre sur le marché des biens et services est donnée par $Y = 850 - 2500\ i$ et l'équation d'équilibre sur le marché de la monnaie par $Y = -500 + 5\ m + 1000\ i$, où Y est le PIB, i est le taux d'intérêt et m l'offre réelle de monnaie ($m = \frac{M}{P}$, M étant la masse monétaire et P l'indice général des prix.

1. On pose $m = 600$. Représentez graphiquement l'équilibre macroéconomique dans le plan iOY . Les droites seront annotées : $IS(1)$ pour l'équilibre sur le marché des biens et services; $LM(1)$ pour le marché de la monnaie. Le point d'équilibre sera marqué par un petit cercle noir et noté $E(1)$. On reportera sur les axes les labels $Y(1)$ et $i(1)$ correspondant aux valeurs d'équilibre du PIB et du taux d'intérêt. Des lignes en pointillés joindront ces valeurs et le point d'équilibre $E(1)$. Le graphe aura un titre et une légende.

2. L'offre de monnaie passe à $m = 800$. Représentez graphiquement la situation sur le même modèle que la question 1 en notant $LM(2)$ la nouvelle droite d'équilibre sur le marché de la monnaie, $E(2)$ le nouvel équilibre, $Y(2)$ la nouvelle valeur d'équilibre du PIB et $i(2)$ le nouveau taux d'intérêt d'équilibre.

3. En superposant les deux précédents graphiques, montrez comment on est passé de $E(1)$ à $E(2)$ par un fléchage suggestif.

4. On admet que la masse monétaire a augmenté régulièrement de $m = 600$ à $m = 800$. Créez une animation montrant simultanément l'évolution de la droite LM et du point d'équilibre.

Solution

Commençons par poser les fonctions exprimant le PIB sur les deux marchés :

```
> restart:
  fis:=i->850-2500*i;#marché des biens et services
```

```
flm:=i->-500+m+1000*i;#marché de la monnaie
```

$$fis := i \rightarrow 850 - 2500\, i$$

$$flm := i \rightarrow -500 + m + 1000\, i \quad \textbf{(5.1)}$$

1. Dans cette question, *m* vaut 600. Maple donne aussitôt la fonction PIB sur le marché de la monnaie pour cette valeur de l'offre réelle de monnaie.

```
> m:=600;
  flm(i);
```

$$m := 600$$

$$100 + 1000\, i \quad \textbf{(5.2)}$$

L'équilibre macroéconomique est le couple (i, Y) vérifiant les deux équations.

```
> solve(fis(i)=flm(i),i);ie(1):=evalf(%);#détermination du
  taux d'intérêt d'équilibre
  Ye(1):=fis(ie(1));#détermination du PIB d'équilibre
```

$$\frac{3}{14}$$

$$ie(1) := 0.2142857143$$

$$Ye(1) := 314.2857142 \quad \textbf{(5.3)}$$

La figure demandée est complexe. Il faut empiler plusieurs graphes et les réunir par la commande **display** du paquetage **plots**.

Le premier graphique trace les droites *IS* et *LM* dans le plan *iOY*. On place déjà sur les axes les valeurs d'équilibre (remarquer la syntaxe de **tickmarks**).

```
> gr1:=plot([fis(i),flm(i)],i=-0.1..0.6,Y=-100..1200,view=[0.
  .0.5,0..1000],labels=["",""],color=[red,blue],tickmarks = [
  [0.1="0.1",ie(1)=typeset(i(1)),0.3="0.3",0.4="0.4",0.5=
  "0.5"],[0="0",200="200",Ye(1)=typeset(Y(1)),400="400",600=
  "600",800="800",1000="1000"]]):gr1;
```



Le second graphique fait appel à **plot** avec l'option **style=line** pour tracer les lignes joignant le point d'équilibre et ses coordonnées sur les deux axes.

```
> gr2:=plot([[0,Ye(1)],[ie(1),Ye(1)],[ie(1),0]],style=line,
```

```
linestyle=dot):gr2;
```



Le troisième graphique se sert de l'option **`style=point`** avec précision sur la couleur demandée dans l'énoncé.

```
> gr3:=plot([[ie(1),Ye(1)]],i=0..0.5,style=point,symbol=
  circle,color=black):gr3;
```



Le quatrième graphique place les annotations demandées. Il faut charger le paquetage plots pour accéder à la commande textplot. Le placement exact des expressions est facilité par l'utilisation du viseur dans le graphe n°1.

```
> with(plots):
  gr4:=textplot({[0.34,103.45,typeset(IS(1))],[0.42,592.41,
  typeset(LM(1))],[0.22,365,typeset(E(1))]}):gr4;
```



Il ne reste plus qu'à réunir les 4 graphiques à l'aide de **`display`**.

```
> G1:=display([gr1,gr2,gr3,gr4]):G1;
```



2. A présent, *m* vaut 800. On commence par récupérer la nouvelle fonction PIB sur le marché de la monnaie.

```
> m:=800;
  flm(i);
```

$$m := 800$$

$$300 + 1000\, i \tag{5.4}$$

L'équilibre macroéconomique est modifié comme suit :

```
> solve(fis(i)=flm(i),i);ie(2):=evalf(%);#détermination du
  taux d'intérêt d'equilibre
  Ye(2):=fis(ie(2));#détermination du PIB d'équilibre
```

$$\frac{11}{70}$$

$$ie(2) := 0.1571428571$$

$$Ye(2) := 457.1428572 \qquad \textbf{(5.5)}$$

On s'inspire de la question précédente pour construire le graphique demandé.

```
> gr1:=plot([fis(i),flm(i)],i=-0.1..0.6,Y=-100..1200,view=[0.
  .0.5,0..1000],labels=["",""],color=[red,blue],tickmarks = [
  [0.1="0.1",ie(2)=typeset(i(2)),0.3="0.3",0.4="0.4",0.5=
  "0.5"],[0="0",200="200",Ye(2)=typeset(Y(2)),400="400",600=
  "600",800="800",1000="1000"]]):
  gr2:=plot([[0,Ye(2)],[ie(2),Ye(2)],[ie(2),0]],style=line,
  linestyle=dot):
  gr3:=plot([[ie(2),Ye(2)]],i=0..0.5,style=point,symbol=
  circle,color=black):
  gr4:=textplot({[0.34,103.45,typeset(IS(1))],[0.42,792.41,
  typeset(LM(2))],[0.17,500,typeset(E(2))]}):
  G2:=display([gr1,gr2,gr3,gr4]):G2;
```



3. Le fléchage est facilité par la commande **arrow** du paquetage **plottools**. On trace une flèche verticale qui suggère le déplacement vers le haut de la droite *LM*.

```
> with(plottools):
  G3:=arrow([0.32, 432], [0.32, 610],0.01,0.04,0.2,
  double_arrow,color=black):display(G3);
```




```
> display({G1,G2,G3},tickmarks = [[0.1="0.1",ie(2)=typeset(i
  (2)),ie(1)=typeset(i(1)),0.3="0.3",0.4="0.4",0.5="0.5"],[0=
  "0",200="200",Ye(1)=typeset(Y(1)),Ye(2)=typeset(Y(2)),400=
  "400",600="600",800="800",1000="1000"]]);
```



4. Cette question invite évidemment à utiliser la commande animate du paquetage plots. On superpose deux graphiques, un pour le déplacement de la droite *LM*, et un pour le déplacement du point d'équilibre.

```
> m:='m';#désassignation de m
  Gr1:=animate(plot,[[fis(i),flm(i)],i=0..0.5],m=600..800,
  color=[red,blue]);Gr1;#déplacement de LM alors que IS reste
  à la même place
```

$$m := m$$

$$Gr1 := PLOT(...)$$



Après avoir calculé les coordonnées du point d'équilibre pour tout *m*, on crée le graphique animé de ce point.

```
> ieq:=solve(fis(i)=flm(i),i);#l'abscisse du point
  d'équilibre est le taux d'intérêt d'équilibre
  Yeq:=fis(ieq);#détermination du PIB d'équilibre
```

$$ieq := \frac{27}{70} - \frac{1}{3500}\,m$$

$$Yeq := -\frac{800}{7} + \frac{5}{7}\,m \qquad \textbf{(5.6)}$$

```
> Gr2:=animate(plot,[[[ieq,Yeq]],i=0..0.5],m=600..800,style=
  point,symbol=circle):Gr2;#graphique pour l'animation du
  point d'équilibre
```




```
> display({Gr1,Gr2},view=[0..0.5,0..1000],labels=[typeset(i),
  typeset(Y)]);#réunion des deux graphiques précédents avec
```

paramétrage des options view et labels



L'animation montre bien que l'accroissement de l'offre de monnaie a pour conséquences une baisse du taux d'intérêt et une hausse du PIB.